Parabolic Subgroups and Induction in `atlas`

Annegret Paul

Western Michigan University

Workshop on the Atlas of Lie Groups and Representations
University of Utah, Salt Lake City
July 10 - 21, 2017

Goals

Tasks

- *Given a real group $G(\mathbb{R})$, a real parabolic subgroup $P(\mathbb{R})$ of $G(\mathbb{R})$ with Levi factor $L(\mathbb{R})$, and $\pi \in \widehat{L(\mathbb{R})}$, use* `atlas` *to find the composition factors of* $\mathrm{Ind}_{P(\mathbb{R})}^{G(\mathbb{R})}(\pi \otimes 1)$.
- *Given a θ-stable parabolic subalgebra $\mathfrak{q} = \mathfrak{l} + \mathfrak{u}$ of $\mathfrak{g}$ and X an irreducible representation of $L(\mathbb{R})$, compute the composition series of $\mathcal{R}_{\mathfrak{q}}^{S}(X)$.*

Application (one of many): Suppose π is unitary (as computed by `atlas`), then all constituents of the induced representation will be unitary as well.

“Real” Example

Example

- $G = GL(4, \mathbb{R})$, $L(\mathbb{R}) = GL(2, \mathbb{R}) \times GL(2, \mathbb{R})$
- $\pi = \sigma_3 \otimes 1$ is the (spherical) three dimensional representation of $GL(2, \mathbb{R})$ with highest weight $(1, -1)$ on the first factor, trivial on the second.
- Consider $Ind_{P(\mathbb{R})}^{G(\mathbb{R})}(\sigma_3 \otimes 1 \otimes 1)$.
- Is this representation reducible? What are the constituents?
- Same questions for $Ind_{P(\mathbb{R})}^{G(\mathbb{R})}(1 \otimes 1 \otimes 1)$.
- More generally, for which values of p and q and $L(\mathbb{R}) = GL(p, \mathbb{R}) \times GL(q, \mathbb{R}) \subset GL(p + q, \mathbb{R})$ is $Ind_{P(\mathbb{R})}^{G(\mathbb{R})}(1 \otimes 1 \otimes 1)$ reducible?

Cohomological Example

Example (Some $A_{\mathfrak{q}}(\lambda)$ Calculations)

Given a real group $G(\mathbb{R})$ and a θ-stable parabolic subalgebra $\mathfrak{q}$ with associated Levi subgroup $L(\mathbb{R})$,

- Construct various $A_{\mathfrak{q}}(\lambda)$ modules.
- How does the result change as we vary λ?
- Reducibility? Unitarity? Non-zero?
- What do the modules and their constituents look like?

Complex Parabolics:

- For a complex group G, types (conjugacy classes) of parabolic subgroups are given by subsets of Π, a fixed set of simple roots.
- In `atlas`, once we define a root datum, a Borel subgroup B (and hence Π) sitting inside our complex group G is fixed.
- There is an `atlas` data type `ComplexParabolic` consisting of a pair (`rd`, `S`). Here `rd` is a root datum, and S is a list of integers corresponding to the numbers (in `atlas` numbering) of the simple roots determining the desired parabolic.

Parabolic Subgroups

Example ($G = GL(4, \mathbb{C})$)

Define the complex parabolic associated to our real parabolic subgroup of $GL(4, \mathbb{R})$ with $L(\mathbb{R}) = GL(2, \mathbb{R}) \times GL(2, \mathbb{R})$:

```
atlas>set G:=GL(4,R)
Variable G: RealForm
atlas> simple_roots (G)
Value:
|  1,  0,  0 |
| -1,  1,  0 |
|  0, -1,  1 |
|  0,  0, -1 |
```

Simple roots are numbered starting at 0; so we need roots 0 and 2.

```
atlas> set rd=root_datum(G)
Variable rd: RootDatum
atlas> rd
Value: simply connected adjoint root datum of Lie type 'A3.T1'
atlas> set Pc=ComplexParabolic:(rd,[0,2])
Variable Pc: (RootDatum,[int])
```

Parabolic Subgroups

For real parabolic subgroups and θ-stable parabolic subalgebras, we need to add an appropriate Cartan involution θ that

- preserves the Levi factor (determined by the subset S of Π); and
- in the real case, for each simple root α that is not in S, $\theta(\alpha)$ is negative;
- in the θ-stable case, for each simple root α that is not in S, $\theta(\alpha)$ is positive.

Parabolic Subgroups

- In `atlas`, the Cartan involution (on the fixed complex group, with **fixed, fixed, fixed** Borel, and now the fixed parabolic subgroup) is specified by a `KGB` element.
- A parabolic subgroup/-algebra can be given by a pair `(S,x)`, where `S` is a list of simple roots, and `x` an appropriate `KGB` element.

Questions

- *How do I know which* `x` *to choose?*
- *When are two parabolics the same (conjugate by K)?*

Parabolic Subgroups

A different way to look at parabolics:

- A conjugacy class of complex parabolic subgroups is G/P for a fixed complex parabolic subgroup P. (Let's call this **fixed** group in `atlas`, uniquely determined by our fixed B and the set of simple roots S, P_S.)
- If θ is the involution corresponding to our chosen `x`, then the K_θ conjugacy class of our parabolic is determined by a K_θ-orbit on G/P_S.
- So parabolic subgroups/-algebras correspond to elements of $K_\theta \backslash G/P_S$.

We need to study these objects $K_\theta \backslash G/P_S$ for a given subset $S \subset \Pi$.

Parabolic Subgroups

- If $\mathtt{x}$ and $\mathtt{y}$ are $\mathtt{KGB}$ elements for the same real form $G(\mathbb{R})$, then there is a bijection/identification (canonical once you have picked a base point: conjugation by an element of G):

$$K_{\theta_x}\backslash G/P_S \leftrightarrow K_{\theta_y}\backslash G/P_S$$

  Call this set just $K\backslash G/P_S$.
- Clearly $B \subset P_S$. Look at $K\backslash G/P_S$ as

$$P_S\text{-orbits on } K\backslash G.$$

- Each element of $K\backslash G/P_S$ is the union of B-orbits. That is, $K\backslash G/P_S$ represents a *partition* of $\mathtt{KGB} = K\backslash G/B$.
- The $\mathtt{KGB}$ elements grouped together are those corresponding to "$\mathtt{KGB}$ for the Levi subgroup".

Parabolic Subgroups

The set S defines an equivalence relation on KGB generated by cross actions and Cayley transforms by roots in S:

$$K\backslash G/P_S = \text{KGB}/\sim_S$$

Example

$G(\mathbb{R}) = GL(4,\mathbb{R})$, $S = [0,2]$. Here is KGB for $G(\mathbb{R})$:

```
0:  0  [C,n,C]   2  0  2    *  1  *  (0,0,0,0)#0 e
1:  1  [C,r,C]   4  1  3    *  *  *  (0,0,0,0) 1 2^e
2:  1  [C,C,C]   0  5  0    *  *  *  (0,0,0,0) 0 1xe
3:  2  [C,C,C]   7  6  1    *  *  *  (0,0,0,0) 1 3x2^e
4:  2  [C,C,C]   1  8  7    *  *  *  (0,0,0,0) 1 1x2^e
5:  2  [n,C,n]   5  2  5    8  *  6  (0,0,0,0) 0 2x1xe
6:  3  [n,C,r]   6  3  6    9  *  *  (0,0,0,0) 1 2x3x2^e
7:  3  [C,n,C]   3  7  4    *  9  *  (0,0,0,0)#1 1x3x2^e
8:  3  [r,C,n]   8  4  8    *  *  9  (0,0,0,0) 1 1^2x1xe
9:  4  [r,r,r]   9  9  9    *  *  *  (0,0,0,0)#2 1^2x3x2^e
```

The KGP orbits are $\{0,2\}$, $\{1,3,4,7\}$, $\{5,6,8,9\}$. The numbers in the sets denote the KGB elements.

Parabolic Subgroups

- In `atlas`, the data type for `KGP` elements is `KGPElt` or `Parabolic`; it consists of a pair `(S,x)`, where S is a list of simple root numbers, and `x` is a `KGB` element.
- We have `(S,x)=(T,y)` iff $S = T$ and $x \sim_S y$.

Parabolic Subgroups

Example (continued)

We can find all `KGP` elements for $G(\mathbb{R})$ and associated to a set S:

```
atlas> KGP(G,[0,2])
Value: [([0,2],KGB element #2),([0,2],KGB element #7),([0,2],
                                                  KGB element #9)]
```

We can find all `KGB` elements in the class of a `Parabolic` P:

```
atlas> set P=KGP(G,[0,2])[2]
Variable P: ([int],KGBElt)
atlas> P
Value: ([0,2],KGB element #9)

atlas> equivalence_class_of (P)
Value: [KGB element #5,KGB element #6,KGB element #8,
                                            KGB element #9]

atlas> set x=KGB(G,6)
Variable x: KGBElt
atlas> P=([0,2],x)
Value: true
```

Parabolic Subgroups

So which `Parabolic` is the one we want (the real parabolic subgroup)? Recall that the involution given by `x`

- preserves the Levi factor (determined by the subset S of Π); and
- in the real case, for each simple root α that is not in S, $\theta_x(\alpha)$ is negative;
- in the θ-stable case, for each simple root α that is not in S, $\theta_x(\alpha)$ is positive.

Each of these conditions applies to one `KGB` element in the class if and only if it applies to all of them.

Parabolic Subgroups

Suppose we have a `KGP` element P so that any element `x` in the class preserves the Levi factor. Then:

- P corresponds to a real parabolic subgroup if and only if the class of `x` contains the maximal element ($\leftrightarrow$ maximally split Cartan).
- P corresponds to a θ-stable parabolic algebra if and only if the class of `x` contains a closed `KGB` orbit.

We can think of those `KGP` elements that correspond to neither (such as `([0,2],KGB element #7)`) as generalized parabolics.

Parabolic Subgroups

`atlas` can tell you what type of parabolic you have, or list, say, all θ-stable parabolics of a real form $G(\mathbb{R})$.

Example (continued)

```
atlas> set kgp=KGP(G,[0,2])
Variable kgp: [([int],KGBElt)]
atlas> is_parabolic_theta_stable (kgp[0])
Value: true

atlas> is_parabolic_real (kgp[0])
Value: false
atlas> is_parabolic_real(kgp[2])
Value: true

atlas> theta_stable_parabolics(G)
Value: [([],KGB element #0),([1],KGB element #1),
                ([0,2],KGB element #2),([0,1,2],KGB element #9)]
```

Levi Subgroups

- Once we have a suitable `KGP` element, it is easy to define the Levi subgroup $L(\mathbb{R}) = MA$.
- `atlas` will define it as it is embedded in $G(\mathbb{R})$. Look at the trivial representation of $L(\mathbb{R})$ to understand the embedding.

Example (continued)

```
atlas> P:=kgp[2]
Value: ([0,2],KGB element #9)
atlas> set L=Levi(P)
Variable L: RealForm
atlas> L
Value: disconnected split real group with Lie algebra
                               'sl(2,R).sl(2,R).gl(1,R).gl(1,R)'


atlas> set t=trivial(L)
Variable t: Param
atlas> t
Value: final parameter(x=3,lambda=[1,-1,1,-1]/2,nu=[1,-1,1,-1]/2)
```

Real Parabolic Induction

There are several induction functions in `atlas`.

- `real_induce_standard` performs real parabolic induction of standard modules; that is, a standard module of $L(\mathbb{R})$ is mapped to a standard module of $G(\mathbb{R})$. Essentially, this is just embedding the parameter, with some ρ shift.
- `real_induce_irreducible` computes the composition series of a representation of $G(\mathbb{R})$ that is induced from an **irreducible** representation on $L(\mathbb{R})$. For this function, all parameters are taken to represent irreducible representations.
- The output is of type `ParamPol`. The function will also accept input of that type.

Real Parabolic Induction

Example ($GL(4,\mathbb{R})$ continued)

First consider $Ind_{P(\mathbb{R})}^{G(\mathbb{R})}(1 \otimes 1 \otimes 1)$:

```
atlas> real_induce_standard (t,G)
Value: non-dominant parameter(x=9,lambda=[3,1,-1,-3]/2,
                                        nu=[1,-1,1,-1]/2)


atlas> real_induce_irreducible(t,G)
Value:
1*parameter(x=9,lambda=[3,1,-1,-3]/2,nu=[1,1,-1,-1]/2) [0]
```

So this induced representation is irreducible. Notice that after the second induction, the parameter is made dominant.

We can define finite dimensional representations by giving their highest weight.

Real Parabolic Induction

Example ($GL(4,\mathbb{R})$ continued)

A representation of $L(\mathbb{R})$ of dimension 3 with highest weight $(1,-1;0,0)$ is

```
atlas> set fd=finite_dimensional(L,[1,-1,0,0])
Variable fd: Param
atlas> fd
Value: final parameter(x=3,lambda=[3,1,1,-1]/2,nu=[3,-3,1,-1]/2)
atlas> dimension(fd)
Value: 3
```

However, this is not the spherical one; it will be the one in the translation family of the trivial representation (the first factor contains the sign representation of $O(2)$).

```
atlas> set fds=parameter(L,3,[1,-1,1,-1]/2,[3,-3,1,-1]/2)
Variable fds: Param
atlas> dimension(fds)
Value: 3
atlas> fd=fds
Value: false
atlas> infinitesimal_character (fd)
Value: [  3, -3,  1, -1 ]/2
atlas> infinitesimal_character (fds)
Value: [  3, -3,  1, -1 ]/2
```

Real Parabolic Induction

Example ($GL(4,\mathbb{R})$ continued)

```
atlas> real_induce_irreducible(fd,G)
Value:
1*parameter(x=9,lambda=[5,1,-1,-1]/2,nu=[3,1,-1,-3]/2) [0]

atlas> real_induce_irreducible(fds,G)
Value:
1*parameter(x=9,lambda=[3,1,-1,-3]/2,nu=[3,1,-1,-3]/2) [0]
1*parameter(x=8,lambda=[3,1,-1,-3]/2,nu=[3,1,-2,-2]/2) [3]
1*parameter(x=6,lambda=[3,1,-1,-3]/2,nu=[2,2,-1,-3]/2) [3]
1*parameter(x=5,lambda=[3,1,-1,-3]/2,nu=[1,1,-1,-1]/1) [4]
```

Real Parabolic Induction

The function will also check whether $L(\mathbb{R})$ is indeed the Levi factor of a **real** parabolic subgroup of $G(\mathbb{R})$:

Example (continued)

Recall that the first `KGP` element in the list `kgp` is θ-stable. Let's try to induce the trivial representation of its Levi factor $GL(2, \mathbb{C})$:

```
atlas> set Q=kgp[0]
Variable Q: ([int],KGBElt)

atlas> set L1=Levi(Q)
Variable L1: RealForm
atlas> L1
Value: connected quasisplit real group with Lie algebra
                                       'sl(2,C).gl(1,C)'

atlas> t:=trivial(L)
Value: final parameter(x=1,lambda=[1,-1,1,-1]/2,nu=[1,-1,1,-1]/2)

atlas> real_induce_irreducible(t,G)
Runtime error:
  L1 is not Levi of real parabolic...
```

Real Parabolic Induction

How does it work?

- $\pi \in \widehat{L(\mathbb{R})}$ is given by a parameter γ, $\pi = J_L(\gamma)$.
- Write $\pi = J_L(\gamma)$ as a formal sum of standard modules for $L(\mathbb{R})$:

$$J_L(\gamma) = \sum_i a_i I_L(\gamma_i), \quad a_i \in \mathbb{Z}.$$

- By Induction by Stages, for each i,

$$I_G(\gamma_i) = Ind_{L(\mathbb{R})}^{G(\mathbb{R})}(I_L(\gamma_i)).$$

Here $I_G(\gamma_i)$ really denotes the output of `real_induce_standard`; the parameter for $L(\mathbb{R})$ is embedded in $G(\mathbb{R})$, with a suitable ρ-shift.

Real Parabolic Induction

- So we have
$$Ind_L^G(J_L(\gamma)) = \sum_i a_i I_G(\gamma_i).$$
- Compute the composition series of each standard module on the right, then the composition series of $Ind_{P(\mathbb{R})}^{G(\mathbb{R})}(\pi \otimes 1)$ (as a formal sum) is obtained as the sum with coefficients a_i.

Theta Stable Parabolic Subalgebras

To define a θ-stable parabolic subalgebra, specify a linear functional λ on a Cartan subalgebra $\mathfrak{t}_0$ of $\mathfrak{k}_0$ taking purely imaginary values. Then

$$\mathfrak{q}(\lambda) = \mathfrak{l}(\lambda) + \mathfrak{u}(\lambda)$$

is given by

$$\Delta(\mathfrak{l}, \mathfrak{t}) = \{\alpha \in \Delta(\mathfrak{g}, \mathfrak{t}) | \langle \lambda, \alpha^\vee \rangle = 0\}$$

$$\Delta(\mathfrak{u}, \mathfrak{t}) = \{\alpha \in \Delta(\mathfrak{g}, \mathfrak{t}) | \langle \lambda, \alpha^\vee \rangle > 0\}$$

- The function `parabolic(lambda,x)` will produce a θ-stable parabolic if `x` is a `KGB` element attached to the fundamental Cartan (and in the distinguished fiber).
- The weight λ must be fixed by the involution θ_x. (The last two conditions are automatic in the equal rank case.)

Theta Stable Parabolic Subalgebras

Example ($Sp(4,\mathbb{R})$)

Let $G(\mathbb{R}) = Sp(4,\mathbb{R})$. We can get a list of all θ-stable parabolic subalgebras:

```
atlas> G:=Sp(4,R)
Value: connected split real group with Lie algebra 'sp(4,R)'
atlas> set tsp=theta_stable_parabolics(G)
Variable tsp: [([int],KGBElt)]
atlas> #tsp
Value: 10

atlas> tsp
Value: [([],KGB element #0),
    ([],KGB element #1),
    ([],KGB element #2),
    ([],KGB element #3),
    ([0],KGB element #2),
    ([0],KGB element #3),
    ([0],KGB element #4),
    ([1],KGB element #5),
    ([1],KGB element #6),
    ([0,1],KGB element #10)]
```

Theta Stable Parabolic Subalgebras

Example ($Sp(4,\mathbb{R})$ continued)

```
atlas> simple_roots(G)
Value:
|  1, 0 |
| -1, 2 |
```

Root 0 is the short root. There are three parabolic subalgebras with the Levi associated to the short root. Which is which?
Let's choose a suitable `KGB` element. If you are used to a system where the first root is compact, choose the `KGB` element accordingly:

```
atlas> print_KGB(G)
...
 0:  0  [n,n]    1   2     4   5  (0,0)#0 e
 1:  0  [n,n]    0   3     4   6  (1,1)#0 e
 2:  0  [c,n]    2   0     *   5  (0,1)#0 e
 3:  0  [c,n]    3   1     *   6  (1,0)#0 e
 ...
```

This will be element 2 or 3.

Theta Stable Parabolic Subalgebras

Example ($Sp(4,\mathbb{R})$ continued)

To define a parabolic with compact $L(\mathbb{R})$, choose $\lambda = (1,1)$, say:

```
atlas> set x=KGB(G,2)
Variable x: KGBElt
atlas> set P=parabolic([1,1],x)
Parabolic is theta-stable.
Variable P: ([int],KGBElt)
atlas> P
Value: ([0],KGB element #2)
atlas> Levi(P)
Value: compact connected real group with Lie algebra 'su(2).u(1)'

atlas> tsp[4]
Value: ([0],KGB element #2)
atlas> tsp[5]
Value: ([0],KGB element #3)
atlas> tsp[6]
Value: ([0],KGB element #4)
```

Example ($Sp(4,\mathbb{R})$ continued)

```
atlas> P:=parabolic([1,-1],x)
Parabolic is theta-stable.
Value: ([0],KGB element #0)
atlas> Levi(P)
Value: connected quasisplit real group with Lie algebra
                                                  'sl(2,R).u(1)'
```

```
atlas> P=tsp[6]
Value: true
```

```
atlas> parabolic([-1,-1],x)
Parabolic is theta-stable.
Value: ([0],KGB element #3)
```

This is the parabolic opposite the first one, with compact Levi factor.

Example ($U(2,2)$)

```
atlas> G:=U(2,2)
Variable G: RealForm
atlas> simple_roots(G)
Value:
|  1,  0,  0 |
| -1,  1,  0 |
|  0, -1,  1 |
|  0,  0, -1 |

atlas> print_KGB(G)
...
 0:  0  [n,n,n]    1   2   3    10   8   6  (0,0,0,0)#0 e
 1:  0  [n,c,n]    0   1   4    10   *   7  (1,1,0,0)#0 e
 2:  0  [c,n,c]    2   0   2     *   8   *  (0,1,1,0)#0 e
 3:  0  [n,c,n]    4   3   0    11   *   6  (0,0,1,1)#0 e
 4:  0  [n,n,n]    3   5   1    11   9   7  (1,1,1,1)#0 e
 5:  0  [c,n,c]    5   4   5     *   9   *  (1,0,0,1)#0 e

 ...
```

If you like to choose $\epsilon_1 - \epsilon_2$ and $\epsilon_3 - \epsilon_4$ to be compact, the correct `element` is either 2 or 5.

Example ($U(2,2)$ continued)

To define the algebra with $L(\mathbb{R}) = U(2,1) \times U(0,1)$:

```
atlas> x:=KGB(G,2)
Value: KGB element #2
atlas> P:=parabolic([1,1,1,0],x)
Parabolic is theta-stable.
Value: ([0,1],KGB element #2)

atlas> L:=Levi(P)
Value: connected quasisplit real group with Lie algebra
                                              'su(2,1).u(1).u(1)'
```

Next Time: θ-Stable Induction.